Variance-stabilizing transformation for parametrized dispersion

This file describes the variance stabilizing transformation (VST) used by DESeq when parametric dispersion estimation is used.
This is a *Mathematica* notebook. The file *vst.pdf* is produced from *vst.nb*.

When using *estimateDispersions* with *fitType="parametric"*, we parametrize the relation between mean μ and dispersion α with two constants a_0 and $a_1$as follows:

```
In[1]:= α = a0 + a1 / μ
```

$$\text{Out[1]}= a_0 + \frac{a_1}{\mu}$$

In the package, a_0 is called the *asymptotic dispersion* and a_1 the *extra-Poisson factor*.

The variance is hence

```
In[2]:= v = μ + α μ^2 // Expand
```

$$\text{Out[2]}= \mu + \mu^2\, a_0 + \mu\, a_1$$

A variance stabilizing transformation (VST) is a transformation u, such that, if X is a random variable with variance-mean relation v, i.e.,$\mathrm{Var}(X) = v(\mathrm{E}(X))$, then $u(X)$ has stabilized variance, i.e., is homoskedastic.

A VST u can be derived from a variance-mean relation v by $u(x) = \int^x \frac{d\mu}{\sqrt{v(\mu)}}$.

Hence, we can get a general VST with

```
In[3]:= u0 = Integrate[ 1/√v, {μ, 0, x}, Assumptions → {a0 > 0, a1 > 0, x > 0}]
```

$$\text{Out[3]}= \frac{\text{Log}\left[\frac{1+2\,x\,a_0+a_1+2\sqrt{x\,a_0\,(1+x\,a_0+a_1)}}{1+a_1}\right]}{\sqrt{a_0}}$$

If u_0 is a VST, then so is $u(x) = \eta\, u_0(x) + \xi$. Hence, this here is a VST, too:

```
In[4]:= u = η u0 + ξ
```

$$\text{Out[4]}= \xi + \frac{\eta\,\text{Log}\left[\frac{1+2\,x\,a_0+a_1+2\sqrt{x\,a_0\,(1+x\,a_0+a_1)}}{1+a_1}\right]}{\sqrt{a_0}}$$

We will now choose the parameters η and ξ such that our VST behaves like $\log_2$ for large values. Let us first look at the asymptotic ratio of the two transformations:

```
In[5]:= Limit[u / Log[2, x], x → ∞, Assumptions → {a0 > 0, a1 > 0, x > 0}]
```

$$\text{Out[5]}= \frac{\eta\,\text{Log}[2]}{\sqrt{a_0}}$$

Hence, if we set η as follows, both tranformations have asymptotically the ratio 1.

In[6]:=
$$\eta = \frac{\sqrt{a_0}}{\text{Log}[2]}$$

Out[6]=
$$\frac{\sqrt{a_0}}{\text{Log}[2]}$$

We also want the difference to vanish for large values:

In[7]:=
```
Limit[u - Log[2, x], x → ∞, Assumptions → {a0 > 0, a1 > 0, x > 0}]
```

Out[7]=
$$\xi + \frac{\text{Log}\left[\frac{4\,a_0}{1+a_1}\right]}{\text{Log}[2]}$$

So, we set

In[8]:=
$$\xi = -\frac{\text{Log}\left[\frac{4\,a_0}{1+a_1}\right]}{\text{Log}[2]}$$

Out[8]=
$$-\frac{\text{Log}\left[\frac{4\,a_0}{1+a_1}\right]}{\text{Log}[2]}$$

Check that both limits are now correct:

In[9]:=
```
Limit[u / Log[2, x], x → ∞, Assumptions → {a0 > 0, a1 > 0, x > 0}]
```

Out[9]=
```
1
```

In[10]:=
```
Limit[u - Log[2, x], x → ∞, Assumptions → {a0 > 0, a1 > 0, x > 0}]
```

Out[10]=
```
0
```

Hence, we arrive at this VST:

In[11]:=
```
FullSimplify[u, Assumptions -> {a0 > 0, a1 > 0, x > 0}]
```

Out[11]=
$$\frac{\text{Log}\left[\frac{1+2\,x\,a_0+a_1+2\sqrt{x\,a_0\,(1+x\,a_0+a_1)}}{4\,a_0}\right]}{\text{Log}[2]}$$

This VST (red) now behaves asymptotically as $\log_2$ (blue), shown here for typical values for a_0 and a_1.

In[12]:=
```
Plot[{u /. {a0 → .01, a1 -> 3}, Log[2, x]}, {x, 0, 10 000}, PlotStyle → {Red, Blue}]
```

Out[12]=



For small values, however, the VST (red) compresses the dynamics much more dramatically than the logarithm (blue) and the identity (green). This reflects that the strong Poisson noise makes differences uninformative for small values.

In[13]:=
```
Plot[{u /. {a0 → .01, a1 -> 3}, Log[2, x], x},
 {x, 0, 100}, PlotStyle → {Red, Blue, Green}, PlotRange → {0, 20}]
```

Out[13]=



A template for the R code in the function:

In[20]:=
```
CForm[FullSimplify[u, Assumptions -> {a0 > 0, a1 > 0, x > 0}]] /.
 {a0 → asymptDisp, a1 → extraPois, x → q}
```

Out[20]//CForm=
```
Log((1 + extraPois + 2*asymptDisp*q +
      2*Sqrt(asymptDisp*q*(1 + extraPois + asymptDisp*q)))/
    (4.*asymptDisp))/Log(2)
```